
Resumen 
Utilidad de las herramientas TIC para el trabajo del docente, dentro y fuera del aula, en concreto, de la 
posibilidad de automatizar el proceso de evaluación con hojas de cálculo. 

1. INTRODUCCIÓN 

Hacemos un paréntesis en nuestra serie de artículos, hasta ahora dedicados a las herramientas on-line 
de Google, para enfocar nuestra atención a la ofimática. Las hojas de cálculo son potentes y, al mismo 
tiempo, sencillos programas que permiten un rápido cálculo basado en fórmulas y parámetros definidos 
por el usuario. Por tanto, son idóneas para ir anotando todas las puntuaciones y evaluaciones de 
nuestros alumnos para, al final, obtener una nota final sin errores y, a ojos de los propios alumnos, 
completamente justa (ya que las notas de todos ellos se calculan con el mismo método). 
Dicho método de cálculo de la nota final puede ser todo lo complicado que nosotros deseemos, y 
basado en múltiples parámetros: actitudes, conocimientos, procedimientos, etc. 
Gracias a la potencia de las hojas de cálculo, podremos también obtener estadísticas y gráficas de los 
resultados. Por supuesto, todo ello se actualiza automáticamente con modificar un solo dato de la hoja 
de cálculo, lo que le da un valor añadido para situaciones del tipo “y si cambio esto, ¿qué pasaría?”. 

En este documento estudiaremos Microsoft Excel 2000, aunque es muy fácil y directo adaptar el 
contenido del mismo a OpenOffice.org2 Calc, como puede verificarse en el apéndice de la página 20. 

2. APLICACIONES PARA EL DOCENTE 

Como se acaba de mencionar, la hoja de cálculo servirá a cualquier docente para ir almacenando todas 
las anotaciones que deseemos sobre nuestros alumnos. En un único fichero podemos tener una hoja 
de cálculo para cada grupo al que impartamos clase, e incluso obtener estadísticas y gráficas globales. 
Dado que podemos introducir cualquier tipo de dato sobre nuestros alumnos, no necesitamos 
restringirnos a notas numéricas, sino que podemos incluir detalles personales como la dirección, 
teléfono, nombre de los padres, observaciones, etc. 
Y aparte de estas aplicaciones, las hojas de cálculo pueden también emplearse en un proceso de 
combinación de correspondencia con un procesador de textos, aprovechando la gran cantidad de datos 
sobre cada alumno que en ellas almacenamos. Pero sobre este tema hablaremos en otro artículo. 

3. FUNCIONAMIENTO 

Está completamente fuera del propósito de este artículo describir el funcionamiento de un programa tan 
completo como Microsoft Excel, por lo que partiremos de unos conocimientos previos que, de no 
tenerse, se pueden adquirir con cualquier libro o tutorial que encontremos en Internet o las librerías. Los 
conceptos que se dan por sabidos son: 

• Libros, hojas, columnas, filas y celdas. 

• Referencias relativas y absolutas. 

• Tipos de datos: textos, numéricos, fechas y fórmulas. 

• Formato y formato condicional de una celda. 

• Es fundamental saber diferenciar entre qué contiene una celda, qué muestra, cómo aparece y con 
qué formato. Por ejemplo: una celda puede contener una fórmula, la cual muestra un resultado 
numérico, el cual aparece con dos decimales y el símbolo del euro, y con un formato de fuente 
‘Arial’ en color rojo. 

• Edición básica: copiar y pegar, seleccionar un rango de celdas, seleccionar una o varias filas, 
seleccionar una o varias columnas, modificar alto de fila y ancho de columna, autosuma, etc. 

• Edición de fórmulas. Funciones básicas (SUMA, PROMEDIO, CONTAR, CONTAR.SI, MIN, MAX, 
REDONEAR, etc.). 

• Uso del controlador de relleno con el botón izquierdo y el derecho del ratón. 

• Creación de gráficas. 

• Personalización de las barras de herramientas. 

Basándonos en estos conocimientos previos, aprenderemos a: 

• Personalizar las barras de herramientas con los botones usados con más frecuencia. 

• Darle un aspecto homogéneo, atractivo y útil a nuestros datos. 

• Agrupar datos. 

• Fórmulas útiles para un docente. 

4. PERSONALIZAR LA BARRA DE HERRAMIENTAS 

Como acabamos de decir, se da por hecho que sabemos mostrar y ocultar barras de herramientas, así 
como añadir, mover o eliminar botones de éstas. Siendo así, deberíamos añadir los siguientes botones 
a la barra de herramientas ‘Formato’: 

•  (Aumentar tamaño de fuente y Disminuir tamaño de fuente). 

•  (Girar texto hacia arriba). 

•  (Celdas). Con este botón entramos en la opción “Celdas” del menú “Formato”, una de las más 
utilizadas. También podemos usar la combinación de teclas Ctrl+1. 

•  (Insertar filas e Insertar columnas). 

•  (Copiar formato). Este botón ya suele aparecen en Excel, y no es necesario añadirlo a la barra 
de herramientas. Permite copiar el formato de una celda a otra(s), no el contenido. Realmente muy 
útil. 

5. INTRODUCIR DATOS Y DARLE ASPECTO A NUESTRA HOJA DE CÁLCULO 

Para empezar, supongamos que impartimos clase a los grupos 3ºESO-D, 4ºESO-C y 1ºBach-B. 
Crearemos en Excel, por lo tanto, un fichero o libro con tres hojas de cálculo: 

Podemos observar que, con el botón derecho del ratón, podemos añadir más hojas de cálculo, o 
eliminar las sobrantes, aparte de cambiarles el nombre. Arrastrando las pestañitas las podemos 
reordenar. 
Ahora dedicaremos la fila 1 para rotular las hojas e identificar las secciones en las que deseemos dividir 
nuestros datos (por ejemplo, “Datos personales”, “1ª Evaluación”, etc.), para lo cual es útil el botón 
“Combinar y centrar”. Como se observa en la ilustración siguiente, daremos un color de fondo a la celda 
A1. A cada grupo le daremos un color diferente, consiguiendo así identificar de un simple vistazo con 
qué grupo estamos trabajando (ejemplo: 3ºD verde, 4ºC azul, 1ºB naranja). 
La fila 2 servirá para dar nombre a los datos que vayamos a introducir; aquí resulta muy útil girar el 
texto 90º, según qué columnas, de forma que éstas ocupen el espacio (ancho) mínimo. Evidentemente, 
añadiremos tantas columnas como nos resulten convenientes, y con un tipo de letra reducido y de fácil 
lectura (Arial Narrow suele ser muy apropiada para este propósito). 

Introduciremos ahora los datos de nuestros alumnos. En cualquier momento podremos insertar nuevas 
columnas en donde nos plazca, con la información que más nos interese. 

Obsérvese que, aplicando tipos de letra, alineaciones, formatos y colores, podemos conseguir un 
aspecto llamativo. En la columna “Años HOY” introduciremos, más adelante, una fórmula que nos 
devuelva los años que tiene hoy cada alumno, basándose en cálculos sobre su fecha de nacimiento y la 
fecha actual. La columna “Repetidor” ha sido definida con un formato condicional, de forma que 
aquellas celdas que contengan la letra ‘S’ se muestran en negrita y de color rojo. 

 
Lo más normal es que en la pantalla no nos quepan todos nuestros alumnos (demasiadas filas) ni 
mucho menos todos los datos de cada uno (demasiadas columnas). Existe un truco que nos permite 
definir las filas y columnas que siempre queremos visibles: arrastrando las barritas que hay junto a las 
barras de desplazamiento. 

 
Gracias a estas barritas, podemos dejar nuestra hoja de cálculo con el siguiente aspecto: 

El único inconveniente es que, ahora, no veremos todas las pestañas de la esquina inferior izquierda 
(las que nos permiten cambiar entre las distintas hojas de cálculo). Pero para estos casos existen, 
precisamente, los cuatro botones que se encuentran justo a la izquierda de las pestañas. 
Ahora podemos crear otra sección, a la derecha de “DATOS PERSONALES”, dedicada a la 1ª 
Evaluación: 

 
Es aconsejable, como se puede ver en la imagen anterior, dejar dos columnas al final llamadas “Nota 
final” y “Evaluación”. En la primera calcularemos mediante una fórmula (más o menos complicada 
según nuestros criterios de evaluación) la nota final de la evaluación, con dos decimales. En la segunda 
redondearemos la cifra anterior sin decimales, que será la nota real que se le asignará al alumno. Esto, 
evidentemente, se podría hacer directamente en una sola columna, pero más adelante veremos un 
práctico método para redondear a nuestro gusto (es decir, que 4’5 siga siendo suspenso y sólo se 
redondee a 5’0 a partir del 4’7, por poner un ejemplo). 
Si el comportamiento del alumno influye en su nota (ya sea restando puntos o porcentualmente), 
deberemos reservar columnas para consignarlo. En la imagen anterior las hemos destacado en rojo.  
También es conveniente hacer cálculos intermedios, de forma que la fórmula para la “Nota final” no sea 
demasiado compleja. Así, nos encontramos con las columnas “Comportam.”, “PRUEBAS” y 
“TRABAJOS”, que representan el total de los subapartados correspondientes. Incluso se puede, como 
aparece en la imagen, hacer aún más cálculos previos: para cada trabajo se evalúa el contenido del 
mismo y su presentación, y en base a esto se calcula la nota del trabajo. La columna “TRABAJOS” 
será, pues, la media de estas notas intermedias. 

6. AGRUPAR DATOS 

Antes de continuar, vamos a conocer una propiedad utilísima (y sencillísima) de Excel, que nos 
permitirá agrupar las columnas que queramos y mostrar u ocultar esos grupos. En nuestro caso 
queremos agrupar los “Datos personales”, y cada una de las tres Evaluaciones. Para ello, 
seleccionamos todas las columnas ocupadas por los datos personales y seleccionamos la opción 
“Datos|Agrupar y esquema|Agrupar”. 

 
El resultado es el siguiente (obsérvese la nueva barra gris que aparece sobre las columnas): 

En el extremo izquierdo de dicha barra, aparecen dos botones, 1 y 2, que nos permiten contraer o 
expandir al mismo tiempo todos los grupos de columnas que haya en la hoja de cálculo. Si sólo 
queremos contraer un grupo concreto, deberemos pulsar el botón – que aparece sobre él (para 
expandir, el botón +). Gracias a esta técnica, a final de curso podríamos mantener oculta la sección de 
datos personales y las de la 1ª y 2ª Evaluaciones, mientras dejamos desplegada sólo la 3ª Evaluación. 
¡Realmente práctico! 

 
Por supuesto, este aspecto invita a colocar, bajo ciertas columnas, fórmulas para calcular la media, el 
mínimo y el máximo de cada una de ellas, como queda ilustrado en la ilustración de la página 10. 

7. FÓRMULAS ÚTILES PARA UN DOCENTE 

Para ilustrar este apartado, partiremos de los siguientes supuestos: 

• El mal comportamiento de un alumno le resta puntos a la nota final, hasta un máximo de 2. 
• La columna “Mala conducta” contiene una cifra que indica el número de veces que se le ha tenido 

que llamar la atención por lo que el profesor considere mala conducta. Por cada una de estas 
veces, el alumno tiene 0’1 puntos menos en la nota final. 

• Cada “Falta injustificada” supone 0’2 puntos menos en la nota final. Una prueba no realizada a 
causa de una ausencia injustificada supone un cero en dicha prueba. 

• Con 11 o más faltas injustificadas, la nota final es un cero. 
• Las tres pruebas realizadas tienen un peso diferente, es decir, no se trata de una media aritmética 

sino ponderada. Los porcentajes serán 30%, 30% y 40%, respectivamente. 

• Para cada trabajo, la presentación será una nota del 0 (mala presentación) al 5 (excelente 

presentación), mientras que el contenido puntuará del 0 al 10. La nota de cada trabajo será la 
media de ambas puntuaciones sobre 10. 

• La nota final será, en un 70% las pruebas, y en un 30% los trabajos, y restándole después lo 
correspondiente a la mala conducta (como ya se ha dicho, un máximo de 2 puntos). 

• En la nota de la evaluación, se redondea hacia arriba a partir de 7 décimas. Es decir, 4’69 se 
redondea a 4, mientras que 4’70 se redondea a 5. Lo mismo en todas las posibilidades: 8’69 es 8 
y 8,70 es 9. 

Supongamos, pues, una hoja de cálculo con los siguientes datos: 

 
Aquí ya aparecen rellenadas no sólo las columnas que el profesor ha evaluado, sino las columnas que 
muestran el resultado de un cálculo, y que habremos de rellenar con fórmulas. Obsérvese el uso del 
formato condicional en las columnas “Faltas injustificadas” (para resaltar los alumnos con 11 o más 
ausencias) y “Evaluación” para resaltar los suspensos de los aprobados. 

7.1. Cómo restringir un valor a un máximo: columna “Comportamiento” 
El contenido de la celda R3 se calcularía, según nuestros supuestos, con la expresión: 

=P3*0,1 + Q3*0,2 
pero, para evitar que el mal comportamiento de un alumno sobrepase el máximo de 2 puntos, la fórmula 
debe ser: 

=MIN(2; P3*0,1 + Q3*0,2) 

7.2. Cómo calcular una media ponderada: columna “PRUEBAS” 
El contenido de la celda V3 debe ser el siguiente (recordemos: 30%, 30% y 40%): 

=S3*0,3 + T3*0,3 + U3*0,4 

7.3. Cómo hacer cálculos con notas que no van de 0 a 10: columnas “Trabajo 1” y “Trabajo 2” 
Sencillamente hay que multiplicar la nota por la factor que permita convertir el rango en otro de 0 a 10. 
En este caso, una nota que va de 0 a 5 simplemente hay que multiplicarla por 2 para que vaya de 0 a 
10. Por lo tanto, el contenido de la celda Y3 debe ser: 

=(W3 + X3*2) / 2 
Aunque Excel nos proporciona una forma más elegante de conseguirlo: 

=PROMEDIO(W3; X3*2) 

7.4. Cómo restringir un valor a un mínimo: columna “NOTA FINAL” 
Si un alumno obtiene una media muy baja, y encima ésta se reduce por mal comportamiento, 
podríamos encontrarnos con el caso de un alumno con nota ¡negativa! La solución es similar a la vista 
más arriba, pero esta vez con la función MAX(). Pero veamos primero cómo hemos de realizar el 
cálculo: 70% las pruebas, 30% los trabajos, y restamos el mal comportamiento. El contenido de la celda 
AD3 debería ser, por tanto: 

= V3*0,7 + AC3*0,3 - R3 
Sin embargo, como acabamos de decir, hemos de evitar notas inferiores a cero: 

=MAX(0; V3*0,7 + AC3*0,3 - R3) 
Pero aún nos queda un cabo suelto: si un alumno acumula 11 o más faltas injustificadas, debe tener un 
cero en la evaluación. Por lo tanto, debemos completar la fórmula anterior con la función condicional 
SI(): 

=SI(Q3>=11; 0; MAX(0; V3*0,7 + AC3*0,3 - R3)) 
No necesita más explicación que ésta: si la celda Q3 (faltas injustificadas) es once o más, la nota será 
0; en caso contrario, la expresión que ya habíamos calculado. 

7.5. Cómo redondear hacia arriba con 7 décimas en lugar de con 5: columna “Evaluación” 
Como todo el mundo sabe, se redondea un número hacia arriba cuando tiene 5 o más décimas, y hacia 
abajo en caso contrario. Pero es muy frecuente que, tratándose de notas finales, el profesor prefiera 
subir el listón, pensando precisamente en los alumnos que rondan el 4’5. 
La función de Excel que realiza los redondeos se llama REDONDEAR(valor; decimales). El primer parámetro 
es el número que queremos redondear, y el segundo parámetro es el número de decimales al que 
deseamos redondear (en nuestro caso, el segundo parámetro será 0, para que resulte un número 
entero). 
El método de cálculo es sorprendente por lo simple que es: sólo tenemos que restar 0’2 antes de 
redondear por el método tradicional. Efectivamente: 

4’69 – 0’2 = 4’49, que al redondear se queda en 4; 
4,70 – 0,2 = 4,50, que al redondear se queda en 5. 

Por lo tanto, el contenido de la celda AE3 será: 
=REDONDEAR(AD3 - 0,2; 0) 

Si quisiéramos redondear a 8 décimas, deberíamos restar 0’3. Y si somos muy magnánimos y vamos a 
aprobar al que tenga un 4’3, deberemos sumar 0’2. 

7.6. Cómo redondear hacia arriba con 7 décimas, pero sólo para aprobar o suspender 
Es decir: 4’69 es suspenso y 4’7 aprobado, pero con el resto de notas redondearemos al estilo 
tradicional (hacia arriba con 5 décimas o más). No nos queda más remedio que utilizar las funciones 
SI() e Y(). 

=SI( Y(AD3>4; AD3<5); REDONDAR(AD3 – 0,2; 0); REDONDEAR(AD3; 0)) 

7.7. Cómo calcular la edad actual de un alumno: columna “Años HOY” 
La celda D3 (véase imagen en pág. 8) debe realizar el cálculo basándose en la fecha de nacimiento del 
alumno (celda C3) y en la fecha actual (función HOY()). Es complicado explicar la fórmula resultante, 
pues combina las funciones Y() y O(), por lo que dejamos su investigación y aclaración a cargo del 
interesado lector: 

= AÑO(HOY()) - AÑO(C3) - SI( O( MES(HOY())<MES(C3); Y(MES(HOY())=MES(C3); 
DIA(HOY())<DIA(C3)));1;0) 

Si no conocemos la fecha de nacimiento de algún alumno, y dejamos dicha celda en blanco, la fórmula 
anterior dará error, por lo que es conveniente adaptarla como sigue: 

=SI(ESBLANCO(C3); ”?”; AÑO(HOY()) - AÑO(C3) 
- SI( O( MES(HOY())<MES(C3); 

Y(MES(HOY())=MES(C3); DIA(HOY())<DIA(C3)));1;0)) 

A esta columna podríamos darle un formato condicional, de forma que quedaran resaltados los alumnos 
con 16 años o más. Así distinguiríamos rápidamente aquellos que no tienen obligación de permanecer 
en la Educación Secundaria Obligatoria. 

7.8. Cómo calcular la nota de un alumno basándonos en las del resto de los alumnos 
Imaginemos la siguiente situación en Educación Física: 

• Para una prueba de salto de longitud anotamos en una columna los centímetros logrados por cada 
alumno. 

• Deseamos darle un 10 al que haya saltado más centímetros (llamémosle alumno A). 

• Al que haya saltado menos (llamémosle alumno B) deseamos darle una nota que decidiremos en 
función de su esfuerzo. Puede ser un 0, o un 3, o un 5. 

• El resto de los alumnos conseguirá una nota proporcional comprendida entre la conseguida por el 
alumno B y la conseguida por el alumno A. 

Dado que ya disponemos de los cálculos mínimo y máximo para cada columna (en las filas 13 y 14), 
sólo nos queda consignar la nota para el alumno B en otra celda (V16): 

 
Efectivamente, vemos que en la columna W, el alumno que menos ha saltado (Ana Carmona Garrido, 
con 152 cm) es la que tiene la nota mínima definida en la celda V16 (un 3); y el alumno que más ha 
saltado (Darío Fernández Fernández, con 246 cm) es el que tiene un 10. ¿Cómo se realiza este 
cálculo? A continuación, la fórmula de la celda W3: 

=(V3-V$13)*(10-V$16)/(V$14-V$13) + V$16 

Obsérvese el uso de referencias absolutas, para facilitar así el arrastre con el controlador de relleno de 
Excel. 
En la imagen anterior no se ha modificado la fórmula de la columna X, por lo que la prueba de longitud 
en realidad no se ha tenido en cuenta para la nota final. Evidentemente, habría que haberla retocado. 
Del mismo modo, el lector debe entender que el caso aquí retratado no es más que un ejemplo, que se 
puede complicar aplicando un criterio diferente para chicos y para chicas. El profesor es, como siempre, 
el que decide cómo desea evaluar a sus alumnos y calcular sus notas. 

7.9. Cómo calcular el número de aprobados y suspensos del grupo 
Debido a los redondeos finales, deberemos fijarnos en la columna AG (“Evaluación”) y no en la AF 
(“NOTA FINAL”) para hacer este cálculo que, por otra parte, es muy sencillo. 

 
El contenido de la celda AG13 (nº de suspensos) será: 

=CONTAR.SI(AG3:AG12; ”<5”) 
Y el de la celda AG14 (nº de aprobados) será: 

=CONTAR.SI(AG3:AG12; ”>=5”) 

O también: 

=CONTAR(AG3:AG12) – AG13 

7.10. Cómo realizar cálculos con notas no numéricas 
En muchos casos puede interesar puntuar a los alumnos en una escala diferente a la usual 0-10. Por 
ejemplo, MM/M/R/B/MB (muy mal/mal/regular/bien/muy bien). Excel nos proporciona una forma muy 
comoda para asignarle, a cada una de estas notas, un valor numérico, de forma que podamos realizar 
cálculos con ellas. 
Lo primer que haremos será, en alguna parte de la hoja de cálculo (o en una hoja de cálculo nueva que 
insertemos en el mismo libro), una tabla de equivalencias. Por ejemplo: 

 
Para facilitar el trabajo, asignaremos a cada una de estas dos tablas de equivalencias un nombre. Esto 
se consigue seleccionando el rango (en la ilustración anterior, se trata del rango “A4:B8” de la hoja 
“Global”) y especificando un nombre en el “Cuadro de nombres” (en la ilustración anterior hemos 
especificado “notas5”). A partir de ahora, podemos referirnos a ese rango con “notas5” en lugar de con 
“Global!$A$4:$B$8”, lo cual es claramente más cómodo. El rango A11:B13 podríamos llamarlo “notas3”. 

Ahora, supongamos que insertamos en la hoja “3ºESO-D” una nueva columna para un tercer trabajo: 

 
Si la celda AF3 contenía el valor: 

=PROMEDIO(AA3; AD3) 
ahora habrá que actualizarla para que tenga en cuenta también el valor de AE3: 

=PROMEDIO(AA3;AD3; BUSCARV(AE3; notas5; 2; 0)) 
La función BUSCARV() busca el primer parámetro (celda AE3) en la primera columna de la tabla 
representada por el segundo parámetro (notas5) y, una vez encontrado, devuelve el valor de la columna 
especificada en el tercer parámetro (2ª columna de notas5). El cuarto parámetro indica el tipo de 
búsqueda realizada; a nosotros nos interesa el tipo 0 (búsqueda exacta). 

8. GRÁFICAS ÚTILES 

8.1. Resultados de una evaluación 
Vamos a definir una gráfica que represente la distribución de las notas de nuestros alumnos, gracias a 
la cual podremos conocer de un vistazo el rendimiento global del grupo (siguiente ilustración). 
Obsérvese que la gráfica deberemos colocarla en las columnas destinadas a la 1ª Evaluación, de forma 
que desaparezca, junto con el resto de los datos, cuando ocultemos dicho grupo. Fijémonos también en 
que necesitamos ciertos datos adicionales (rango AF18:AG28), fácilmente calculables, sobre los que 
basar la gráfica. 

En concreto, las celdas AF18:AF28 contienen los números del 0 al 10. Las celdas AG18:AG28 
contienen el número de alumnos que ha obtenido cada una de esas notas en la evaluación. Lo cual se 
calcula con la siguiente fórmula (para la celda AG18): 

=CONTAR.SI(AG$3:AG$12; AF18) 
Como ya dijimos al principio, no vamos a explicar cómo se crean gráficas; simplemente diremos qué 
parámetros tendría la gráfica de la ilustración anterior: 

• Tipo de gráfica: Columnas. 
• Serie: 

• Nombre: Resultados de la 1ª Evaluación 
• Valores: $AG$18:$AG$28” 
• Rótulos del eje de categorías (X): $AF$18:$AF$28. 

8.2. Gráfica comparativa entre grupos 

En nuestro libro de Excel de ejemplo disponemos, como ya sabemos, de una hoja de cálculo por cada 
uno de los grupos a los que impartimos clase. Podemos añadir una hoja de cálculo adicional donde 
realizar estadísticas y gráficas conjuntas de todos los grupos. Por ejemplo, podríamos crear una gráfica 
de áreas o de columnas donde se visualizaran los resultados de la 1ª Evaluación, especificando una 
serie por cada grupo. Es decir, exactamente igual que en el epígrafe anterior, pero introduciendo en la 
misma gráfica una serie por cada grupo. 

8.3. Gráfica de la evolución de los alumnos 
Cuando ya están los datos de las tres evaluaciones, podemos incluir una gráfica (a la derecha de la 3ª 
Evaluación), del tipo Líneas, donde se muestre la evolución de las notas finales de cada alumno. El 
resultado podría quedar así: 

Evolución de los alumnos

10

1ª 2ª 3ª Final

Evaluación

N
ot

a

Bermúdez López, José Manuel

Carmona Garrido, Ana

Delgado Torres, Guadalupe

Díaz Rodríguez, Samuel

Fernández Fernández, Darío

Fernández Matilla, José Mª

Fernández Sánchez, Sergio

Martínez Bolív ar, Pedro

Moraleda Zafra, Neomí

Trav esí Gómez, Manuel Jesús

Carmona Garrido, Ana

Delgado Torres, Guadalupe

Díaz Rodríguez, Samuel

Fernández Fernández, Darío

Fernández Matilla, José Mª

Fernández Sánchez, Sergio

Martínez Bolív ar, Pedro

Moraleda Zafra, Neomí

Trav esí Gómez, Manuel Jesús
 

El truco para conseguir este tipo de gráficas es muy fácil, ya que Excel permite usar datos no contiguos 
para una misma gráfica. ¿Cómo? Al insertar el gráfico, escogemos el tipo Líneas: 

 
En el siguiente paso, marcamos “Series en Filas” y pulsamos el botón “Rango de datos”: 

Ahora seleccionamos el rango con los nombres de los alumnos, incluyendo el título de esa 
columna (es decir, el rango B2:B12), como muestra la ilustración de la izquierda; luego, 
pulsando la tecla Ctrl, vamos añadiendo los rangos correspondientes a las notas de todas 
las evaluaciones, como muestra la ilustración de la derecha: 

               
Cuando estén todos los rangos seleccionados, volvemos a la configuración de la gráfica. 
Excel automáticamente identificará los rótulos de los ejes, las series, las categorías, etc. 
Sólo nos quedará darle el aspecto que nosotros queramos, hasta conseguir algo parecido a 
lo mostrado antes, en la página 18. 

9. APÉNDICE: OPENOFFICE.ORG2 CALC 

Si hay una aplicación de OpenOffice.org casi idéntica a su equivalente en Microsoft Office, ésa es Calc. 
Efectivamente, todo lo aquí expuesto para Excel se puede realizar exactamente igual en Calc, 
simplemente utilizando la siguiente traducción de las funciones (los parámetros son los mismos y en el 
mismo orden): 

Excel Calc  
CONTAR() COUNT() 
CONTAR.SI() COUNTIF() 
PROMEDIO() AVERAGE() 
REDONDEAR() ROUND() 
SI() IF() 

Aunque Excel supera sobradamente a Calc en la funcionalidad del controlador de relleno (se puede 
arrastrar con el botón derecho del ratón, lo cual es utilísimo), Calc supera a Excel en el número de 
funciones disponibles para usar en las fórmulas. Efectivamente, en Calc no es necesario utilizar una 
complicada fórmula para calcular los años de un alumno (véase pág. 12); simplemente deberemos 
usar la siguiente fórmula: 

=YEARS(C3; TODAY(); 0) 
La función YEARS() calcula, precisamente, la diferencia en años entre dos fechas, y la función 
TODAY() devuelve la fecha actual. El tercer parámetro puede ser 0 o 1. Si es 0, calcula la diferencia 
teniendo en cuenta el día y mes de ambas fechas; si es 1, calcula la diferencia fijándose sólo en el año 
de ambas fechas. 

Autoría 

 Nombre y Apellidos: Ignacio Triviño Mosquera 
 Centro, localidad, provincia: I.E.S. “Montes Orientales”, Iznalloz, Granada 
 E-mail: itrivi@gmail.com 

